Slides for Chapter 11: Basics of Dynamic Modeling

11.1 Comparative steady-state analysis

As in the case of labor supply, we would like to have models in which the stock of capital is endogenous. I will present a shortcut that is very valuable in many situations.

In models for which there exists a steady state, it is possible to represent this steady state as a complementarity problem.

We can then at least perform comparative steady-state experiments, in which a parameter change moves us from one steady state to another.

Model 11-1 incorporates optimal capital accumulation via the use of a rationing constraint and endogenous “taxes” to create a model for comparative steady state analysis when the capital stock is endogenously adjusted to its steady-state value.

Let r denote the rental price (for one period) of a unit of capital and let p_k denote the price of a new unit of capital.

δ will denote the rate of depreciation of capital per period, and ρ will denote the discount rate between periods.

The steady-state optimal capital accumulation condition is a relationship between the price and rental rate on capital. The rental rate must be given by:

$$r = \left[1 - \frac{1 - \delta}{1 + \rho} \right] p_k$$

The rental price (r) is equal to the price of creating a new unit of capital (p_k) minus the present value of what the capital could be sold for next period.

In the steady state, the price of a new unit next period is the same, so an old unit can be sold for its original value minus one period's depreciation (1 - δ).

The present value of the undepreciated portion is thus (1 - δ)/(1+ ρ).

The dynamic steady-state problem is represented as a static problem using two tricks.

First, the use of capital is subsidized to create the desired wedge (denoted TAU) between the rental price and the price of new capital

```
TAU = - (1 - DELTA) / (1 + RHO);
```

This then appears as a subsidy to capital use, creating the wedge between the rental price and the price of producing a new unit of capital (PK).

```
PRF_X.. 100 * PL**0.4 * (PK*(1+TAU)/0.5)**0.6 =G= 100 * PX;

PRF_Y.. 100 * PL**0.6 * (PK*(1+TAU)/0.5)**0.4 =G= 100 * PY;
```

One unit of new capital is produced using one unit of labor.

```
PRF_K..60*PL =G= 60 * PK * (1-KTAX);
```

Second, since in the steady state newly produced capital is equal to depreciation and depreciation is equal to a share DELTA of total capital,

an auxiliary constraint is specified to "endow" consumers with the carryforward from the previous period.

Let K_s denote the capital stock and K_n the production of new capital. In the steady state, these are related by:

$$K_n = \delta K_s \qquad K_s = \frac{K_n}{\delta}$$

The carryforward from the previous period is the steady state stock minus new production. It is is called KFORWD and given by:

carry forward = *KFORWD* =

$$K_s - K_n = \frac{K_n}{\delta} - K_n = \frac{(1-\delta)K_n}{\delta}$$

We will give the consumer an endowment, via the rationing multiplier KFORWD, the quantity of capital on the right-hand side of the above equation.

```
A_KFORWRD.. 140*KFORWRD  =E= 60*K * (1-DELTA) / DELTA;
```

Parameters: $\rho = 0.4$, $\delta = 0.3$ imply $\tau = 0.5$, $r = 0.5 * p_k$

Markets	Production Sectors X	Y	K	W	Consumers CONS
PX	100			-100	
PY		100		-100	
PW				200	-200
PL	- 40	-60	-60		160
PK	-120	-80	60		140
SUB	60	40			-100

Carryforward:	140
New capital:	60
Total capital available for production:	200

Carryforward = 140 = $60*(1 - \delta) / \delta$ where $\delta = 0.3$

Price of capital chosen to be equal to one.

User cost of capital in X: 120 units of K at a price of 0.5
User cost of capital in Y: 80 units of K at a price of 0.5

Counterfactuals:

(1) raise the rate of time preference to $\rho = 0.6$

(3) restore ρ to 0.4 and set a tax on new capital production to 0.20

```
$TITLE: M11-1.GMS: steady state capital stock, comparative steady-states


$ONTEXT
 "closure rule": instead of the capital stock being fixed
 (quantity closure), the stock
 adjusts to satisfy the steady-state relationship between the rental
 rate and the price of producing new capital (price closure):
   delta = depreciation rate
   rho   = rate of time preference

 rental rate = (1 - (1 - delta)/(1 + rho))*(price of new capital)

 this is done via a subsidy to capital use that creates the wedge
   subsidy = (1 - delta)/(1 + rho)

 KFORWARD is (undepreciated) capital taken forward from previous
   period and K is new capital production, which can be used instantly.

 The steady-state condition is depreciation equal new capital formation
  (KFORWARD + K)*DELTA = K   or   KFOWARD = K*(1-DELTA)/DELTA
```

```
                 Production Sectors          Consumers

   Markets    |    X        Y     K      W     |        CONS
   ------------------------------------------------------
        PX    |   100                -100      |
        PY    |            100       -100      |
        PW    |                       200      |        -200
        PL    |  - 40      -60   -60           |         160
        PK    |  -120      -80    60           |         140
        SUB   |    60       40                 |        -100
   ------------------------------------------------------
$OFFTEXT

PARAMETERS
   RHO          Time preference parameter
   DELTA        Depreciation rate
   TAU          Effective capital use subsidy
   KTAX         Tax on new capital production
   NEWCAP       New capital stock after counterfactual (= 1 initially);

RHO     = 0.4;
DELTA   = 0.3;
TAU     = - (1 - DELTA)/(1 + RHO);
KTAX    = 0;

NONNEGATIVE VARIABLES
```

```
    X       Activity level for sector X
    Y       Activity level for sector Y
    W       Activity level for sector W (Hicksian welfare index)
    K       Capital stock index

    PX      Price index for commodity X
    PY      Price index for commodity Y
    PL      Price index for primary factor L
    PK      Price index for primary factor K
    PW      Price index for welfare (expenditure function)

    CONS    Income definition for CONS
    KFORWRD Capital stock from previous period;

EQUATIONS
    PRF_X   Zero profit for sector X
    PRF_Y   Zero profit for sector Y
    PRF_W   Zero profit for sector W (Hicksian welfare index)
    PRF_K   Zero profit for capital index

    MKT_X   Supply-demand balance for commodity X
    MKT_Y   Supply-demand balance for commodity Y
    MKT_L   Supply-demand balance for primary factor L
    MKT_K   Supply-demand balance for factor K
    MKT_W   Supply-demand balance for aggregate demand
```

```
   I_CONS   Income definition for CONS
   A_KFORWRD  Auxiliary equation to determine the carry forward;

*       Zero profit conditions:

PRF_X..    100 * PL**0.4 * (PK*(1+TAU)/0.5)**0.6 =G= 100 * PX;

PRF_Y..    100 * PL**0.6 * (PK*(1+TAU)/0.5)**0.4 =G= 100 * PY;

PRF_W..    200 * PX**0.5 * PY**0.5 =G= 200 * PW;

PRF_K..    60*PL =G= 60 * PK * (1-KTAX);

*       Market clearing conditions:

MKT_X..    100 * X =G= 100 * W * PW/PX;

MKT_Y..    100 * Y =G= 100 * W * PW/PY;

MKT_W..    200 * W =G= CONS / PW;

MKT_L..    160  =G= 60*K + 40* X*PX/PL + 60*Y*PY/PL;

MKT_K..    140*KFORWRD + 60*K =G= 120 * X * PX*0.5/(PK*(1+TAU)) +
                                   80 * Y * PY*0.5/(PK*(1+TAU));
```

```
*       Income constraints:

I_CONS..   CONS =E= 160 * PL + 140*KFORWRD *PK +
                   PK * TAU * (120 * X * PX*0.5/(PK*(1+TAU)) +
                                80 * Y * PY*0.5/(PK*(1+TAU))) +
                                60* PK * KTAX * K;

*       Auxiliary constraints:

A_KFORWRD.. 140*KFORWRD  =E= 60*K * (1-DELTA) / DELTA;

MODEL ALGEBRAIC /PRF_X.X, PRF_Y.Y, PRF_W.W, PRF_K.K,
                 MKT_X.PX, MKT_Y.PY, MKT_L.PL, MKT_K.PK, MKT_W.PW,
                 I_CONS.CONS, A_KFORWRD.KFORWRD/;

X.L     =1;
Y.L     =1;
W.L     =1;
K.L     =1;

PX.L    =1;
PY.L    =1;
PK.L    =1;
PW.FX   =1;
PL.L    =1;
```

```
CONS.L  =200;

KFORWRD.L = 1;

ALGEBRAIC.ITERLIM = 0;
SOLVE ALGEBRAIC USING MCP;
ALGEBRAIC.ITERLIM = 2000;
SOLVE ALGEBRAIC USING MCP;

*  Raise the rate of time preference from 0.4 to 0.6:

RHO  = 0.6;
TAU  = - (1 - DELTA)/(1 + RHO);

SOLVE ALGEBRAIC USING MCP;

NEWCAP = K.L;
DISPLAY NEWCAP;

*  Set rho back to 0.4, tax new capital at 0.20

RHO  = 0.4;
TAU  = - (1 - DELTA)/(1 + RHO);

KTAX = 0.20;
```

```
SOLVE ALGEBRAIC USING MCP;

NEWCAP = K.L;
DISPLAY NEWCAP;
```

11.2 Converting an Infinite-Horizon Problem to an MCP

Comparative dynamics in a model with a steady-state

Consider a straight-forward dynamic model with an endogenous capital stock.

Sectors (Activities)

X_t	production of composite good in period t
I_t	production of new capital (investment) in period t
K_t	transforms capital into capital services and future capital

Commodities (Markets)

px (CXt) price of X in period t
pr_t (CRt) rental price of capital in period t
pk_t (CKt) asset price of capital (price of a new capital good) in period t
pl_t (CLt) price of labor in period t

Consumers

Infinitely lived representative consumer

δ = rate of capital depreciation
ρ = rate of time preference (discounting utility)
KE_t = capital endowment at the beginning of a period
K_t = capital stock for production at time t ($KE_t + I_t$)

Conditions for Steady-State Equilibrium:

(1) $$KE_{t+1} = KE_t \quad \Rightarrow \quad I_t = \delta KE_t$$

(2) rate of interest = ρ: $$pj_{t+1} = \frac{pj_t}{1+\rho} \quad \textit{for all goods } j = X,I,K$$

(3) $$pr_t = \left[1 - \frac{1-\delta}{1+\rho}\right] pk_t$$ relationship between asset and rental prices

(4) $$KE_{t+1} = (1-\delta)K_t = (1-\delta)(KE_t + I_t)$$ (KE: capital endowment)

$$pk_{t+1}KE_{t+1} = (1-\delta)\frac{pk_t}{1+\rho}K_t = (1-\delta)\frac{(pk_t KE_t + pk_t I_t)}{1+\rho}$$

	Xt	It	Kt	CONS	
CXt	200			-200	0
CRt	-100		100[3]		0
CKt		40	-400[2]	360[1]	0
CLt	-100	-40		140	0
CKt+1			300[4]	-300	0

Parameters: RHO = 0.2, DELTA = 0.1

Prices: CX0=CR0=CL0=1: CK0=4 CK1=CK0/(1+RHO)= 3.3333

CR0 = (1 - (1-DELTA)/(1+RHO))*CK0 = (1/4)*CK0

1   360 = 90 units at CK0 = 4
2   400 = 100 units at CK0 = 4
3   100 = 100 units at rental price = 1
4   300 = undepreciated capital (1-delta)*100 = 90
at a price of CK1 = 1/(1+RHO) = 4/1.2 = 3.3333
300 = (1-DELTA)*4*100/(1+RHO)= 300

The amount 360 - 300 = 60 can be thought of as net rental income:

rental income (90) minus the cost of replacing depreciated capital: 9*CK0/(1+RHO) = 30.

Problem: Suppose we want to represent this infinite-horizon problem as a finite dimension complementarity problem

Approaching the last period the consumer would have no incentive to accumulate capital and would want to run down the capital stock.

(1) Assume a finite number of periods plus a terminal period.

(2) Assume an extra dummy agent: DEITY

(3) Assume that the dummy agent is endowed with an extra good HEAVEN

(4) Assume that the dummy agent will only sell Heaven in exchange for terminal period capital (does not demand any other good)

(5) Assume that the representative agent has a demand for heaven

(6) Use a tax/subsidy on heaven to ensure that the asset/rental price relationship holds on terminal capital (so that the economy is forced onto the steady-state path at terminal time)

Terminal period

	Xt	It	Kt	CONS	DEITY	
CXt	200			-200	0	0
CRt	-100		100		0	0
CKt		40	-400	360	0	0
CLt	-100	-40		140	0	0
CKt+1			300		-300	0
Heaven				-300	300	0
	0	0	0	0		

```
$TITLE: M11-2.GMS:  Infinite horizon dynamic model, MPS/GE formulation

$ONTEXT
Converts infinite horizon model in to a fixed time horizon MCP
Trick is that there is an outside agent "DEITY" who demands terminal
period capital in exchange for the good "HEAVEN"
Price of HEAVEN relative to terminal capital is set by a tax/subsidy
so that the steady-state condition is satisfied in the terminal period
$OFFTEXT

SETS T  time periods /1*25/;

PARAMETERS
 DELTA          rate of depreciation
 RHO            rate of time preference
 PV             present value from terminal period to first period
 TERM           number of the terminal period (25)
 RTERM          present value of terminal values at t = 1
 INITK          initial capital stock
 R(T)           present value of period t values at t = 1
 D(T)           remaining undepreciated portion of initial K at t
 PVUTIL         present value of the utility path at t = 1
 TLAST(T)       switch to indicate the terminal (last) period
 TFIRST(T)      switch to indicate the first time period
 SOLUTION(T,*) stores values of the solution path
 CONSUME(T)     consumption at time t
```

```
 INVEST(T)     investment at time t
 KSTOCK(T)     capital stock at time t;


RHO = 0.2;
DELTA = 0.1;
INITK = 90;
TERM = CARD(T);
RTERM = (1/(1+RHO))**(CARD(T) - 1);;
R(T) = (1/(1+RHO))**(ORD(T)-1);
D(T) = (1-DELTA)**(ORD(T) - 1);
PV = 200*SUM(T, R(T)) + 90*(4*RTERM/(1+RHO));
TLAST(T) = 0;
TLAST('25') = 1;
TFIRST('1') = 1;


$ONTEXT
$MODEL: BASIC

$SECTORS:
 X(T)     ! production at time t
 I(T)     ! production of new capital (investment) at time t
 K(T)     ! capital stock at time t
 U        ! present value of utility
```

```
$COMMODITIES:
 CX(T)    ! present value price of output at t
 CR(T)    ! present value rental rate for capital at time t
 CK(T)    ! present value price of capital (cost of production) at t
 CL(T)    ! present value price of labor at time t
 CKT      ! terminal period present value price of capital
 CU       ! price of utility (intertemporal consummer price index)
 HEAVEN   ! price of heaven

$CONSUMERS:
 CONS     ! representative consumer
 DEITY    ! deity who demand terminal capital stock and sells heaven

$AUXILIARY:
 TRANS    ! endogenous tax or subsidy achieves steady-state at TLAST

$PROD:K(T)
        O:CK(T+1)        Q:(100*(1-DELTA))   P:(4*R(T+1))
        O:CKT$TLAST(T)   Q:(100*(1-DELTA))   P:(4*R(T)/(1+RHO))
        O:CR(T)                 Q:100        P:(R(T))
        I:CK(T)                 Q:100        P:(4*R(T))

$PROD:I(T)
        O:CK(T)          Q:10
        I:CL(T)          Q:40
```

```
$PROD:X(T) s:1
        O:CX(T)         Q:200
        I:CL(T)         Q:100
        I:CR(T)         Q:100

$PROD:U  s:1 a:2
        O:CU      Q:PV
        I:CX(T)   Q:200    P:R(T)  a:
        I:HEAVEN  Q:90     P:(4*RTERM/(1+RHO))   A:CONS  N:TRANS

$DEMAND:CONS
        D:CU                   Q:PV
        E:CL(T)                Q:140
        E:CK(T)$TFIRST(T)  Q:INITK

$DEMAND:DEITY
        D:CKT           Q:90
        E:HEAVEN        Q:90

$CONSTRAINT: TRANS
 CR('25') =E= (1 - (1-DELTA)/(1+RHO))*CL('25')*4;


$OFFTEXT
$SYSINCLUDE MPSGESET BASIC
```

```
TRANS.UP = +INF;
TRANS.LO = -INF;

CX.L(T) = R(T);
CL.L(T) = R(T);
CR.L(T) = R(T);
CK.L(T) = 4*R(T);
CKT.L = 4*R('25')/(1+RHO);
HEAVEN.L = 4*R('25')/(1+RHO);
TRANS.L = 0;

*BASIC.ITERLIM = 0;

$INCLUDE BASIC.GEN
SOLVE BASIC USING MCP;

PVUTIL = SUM(T, X.L(T)*R(T)) + (X.L('25')*R('25'))/RHO;

DISPLAY PVUTIL;


CONSUME(T) = X.L(T);
INVEST(T) = I.L(T);
KSTOCK(T) = K.L(T);
SOLUTION(T,"X") = X.L(T);
SOLUTION(T,"I") = I.L(T);
```

```
SOLUTION(T,"K") = K.L(T);

$LIBINCLUDE XLDUMP SOLUTION M11.xls SHEET1!A2

* counterfactual: lower the capital stock below is ss value

INITK = 30;

$INCLUDE BASIC.GEN
SOLVE BASIC USING MCP;

PVUTIL = SUM(T, X.L(T)*R(T)) + (X.L('25')*R('25'))/RHO;

DISPLAY PVUTIL;


CONSUME(T) = X.L(T);
INVEST(T) = I.L(T);
KSTOCK(T) = K.L(T);
SOLUTION(T,"X") = X.L(T);
SOLUTION(T,"I") = I.L(T);
SOLUTION(T,"K") = K.L(T);

$LIBINCLUDE XLDUMP SOLUTION M11.xls SHEET1!F2

* counterfactual: lower the rate of time preference
```

```
INITK = 90;

RHO = 0.1;
RTERM = (1/(1+RHO))**(CARD(T) - 1);;
R(T) = (1/(1+RHO))**(ORD(T)-1);
D(T) = (1-DELTA)**(ORD(T) - 1);
PV = 200*SUM(T, R(T)) + 90*(4*RTERM/(1+RHO));

$INCLUDE BASIC.GEN
SOLVE BASIC USING MCP;

PVUTIL = SUM(T, X.L(T)*R(T)) + (X.L('25')*R('25'))/RHO;

DISPLAY PVUTIL;


CONSUME(T) = X.L(T);
INVEST(T) = I.L(T);
KSTOCK(T) = K.L(T);
SOLUTION(T,"X") = X.L(T);
SOLUTION(T,"I") = I.L(T);
SOLUTION(T,"K") = K.L(T);

$LIBINCLUDE XLDUMP SOLUTION M11.xls SHEET1!K2
```

	X	I	K		X	I	K		X	I	K
1	1	1	1	1	0.54686	2.006727	0.500673	1	0.916565	1.499733	1.049973
2	1	1	1	2	0.66269	1.74314	0.624919	2	0.946172	1.445869	1.089563
3	1	1	1	3	0.747622	1.553135	0.717741	3	0.969565	1.403438	1.12095
4	1	1	1	4	0.810585	1.413703	0.787337	4	0.988074	1.369941	1.145849
5	1	1	1	5	0.857555	1.310369	0.83964	5	1.002734	1.343456	1.16561
6	1	1	1	6	0.892732	1.233325	0.879009	6	1.014355	1.32249	1.181298
7	1	1	1	7	0.919145	1.175656	0.908673	7	1.023571	1.305879	1.193756
8	1	1	1	8	0.939013	1.132375	0.931044	8	1.030885	1.292708	1.203651
9	1	1	1	9	0.953976	1.099831	0.947922	9	1.036691	1.282259	1.211512
10	1	1	1	10	0.965255	1.075328	0.960663	10	1.041301	1.273967	1.217758
11	1	1	1	11	0.973763	1.056862	0.970283	11	1.044962	1.267385	1.22272
12	1	1	1	12	0.980183	1.042936	0.977548	12	1.04787	1.262158	1.226664
13	1	1	1	13	0.985031	1.032427	0.983036	13	1.050181	1.258006	1.229798
14	1	1	1	14	0.988691	1.024495	0.987182	14	1.052017	1.25471	1.232289
15	1	1	1	15	0.991455	1.018506	0.990314	15	1.053475	1.252092	1.23427
16	1	1	1	16	0.993544	1.013984	0.992681	16	1.054634	1.250013	1.235844
17	1	1	1	17	0.995121	1.010569	0.99447	17	1.055555	1.248363	1.237096
18	1	1	1	18	0.996313	1.00799	0.995822	18	1.056286	1.247055	1.238092
19	1	1	1	19	0.997213	1.006045	0.996844	19	1.056867	1.24602	1.238885
20	1	1	1	20	0.997893	1.004579	0.997618	20	1.057328	1.245203	1.239516
21	1	1	1	21	0.998406	1.003479	0.998204	21	1.057694	1.244562	1.240021
22	1	1	1	22	0.998793	1.002658	0.998649	22	1.057983	1.244064	1.240425
23	1	1	1	23	0.999084	1.002054	0.99899	23	1.058211	1.243683	1.240751
24	1	1	1	24	0.999302	1.001625	0.999253	24	1.05839	1.243402	1.241016
25	1	1	1	25	0.999462	1.001344	0.999462	25	1.058529	1.243208	1.241235

Destroy 2/3 of initial (steady state) capital stock





Lower rate of time preferencce from 0.2 to 0.1
1.6
1.5
1.4
1.3
1.2
1.1
1
0.9
0.8
1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19 20 21 22 23 24 25
X
I
K